

VSL specification

1 Grammar

program \rightarrow function_list
function_list \rightarrow function | function_list function
statement_list \rightarrow statement | statement_list statement
print_list \rightarrow print_item | print_list print_item
expression_list \rightarrow expression | expression_list ',' expression
variable_list \rightarrow variable | variable_list ',' variable
argument_list \rightarrow expression_list | ϵ
parameter_list \rightarrow variable_list | ϵ
declaration_list \rightarrow declaration_list declaration | ϵ
function \rightarrow FUNC variable '(' parameter_list ')' statement

statement \rightarrow assignment_statement | return_statement | print_statement
| null_statement | if_statement | while_statement | block

block \rightarrow '{' declaration_list statement_list '}'
assignment_statement \rightarrow variable ':' '=' expression
return_statement \rightarrow RETURN expression
print_statement \rightarrow PRINT print_list
null_statement \rightarrow CONTINUE
if_statement \rightarrow IF expression THEN statement FI
| IF expression THEN statement ELSE statement FI
while_statement \rightarrow WHILE expression DO statement DONE

expression \rightarrow expression '+' expression | expression '-' expression |
expression '*' expression | expression '/' expression | '-' expression |
'(' expression ')' | integer | variable | variable '(' argument_list ')'

declaration \rightarrow VAR variable_list
variable \rightarrow letter | variable alphanumeric
integer \rightarrow digit | integer digit
print_item \rightarrow expression | text
text \rightarrow text_character | text_character text
text_character \rightarrow printable_character | escaped_character
printable_character \rightarrow *any printable ASCII character*
escaped_character \rightarrow \n

alphanumeric \rightarrow letter | digit
digit \rightarrow '0' | '1' | \dots | '9'
letter \rightarrow 'a' | 'b' | \dots | 'z'